

```

1  ' -----
2  '       From the book Inside Active Directory, ISBN 0-201-61621-1
3  '       Copyright (C) 2002 by Addison-Wesley
4  '       Script by Sakari Kouti (see http://www.kouti.com)
5  ' You have a royalty-free right to use, modify, reproduce and distribute
6  ' this script (and/or any modified version) in any way you find useful,
7  ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8  ' warranty, obligations or liability for the script. If you modify
9  ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Dim intYesNo, objSchema, objAttr, objClass, objDSE
14
15 intYesNo = MsgBox("Create a new attribute and class?", _
16     vbYesNo + vbQuestion + vbDefaultButton2, _
17     "Extending the Schema")
18 If intYesNo = vbNo Then
19     WScript.Quit(0) 'no error so no errorlevel
20 End If
21
22 Set objSchema = GetObject("LDAP://dc1.sanao.com/" & _
23     "CN=Schema,CN=Configuration,DC=sanao,DC=com")
24
25 On Error Resume Next
26
27 Set objAttr = objSchema.Create("attributeSchema", _
28     "CN=Sanao-Com-2000-MyProduct-User-Data")
29
30 Call objAttr.Put("LDAPDisplayName", _
31     "sanaoCom2000-MyProduct-UserData")
32 Call objAttr.Put("attributeID", "1.3.6.1.4.1.123123123.2.1")
33 Call objAttr.Put("attributeSyntax", "2.5.5.10")
34 Call objAttr.Put("oMSyntax", 4)
35 Call objAttr.Put("isSingleValued", True)
36 Call objAttr.Put("rangeLower", 200)
37 Call objAttr.Put("rangeUpper", 200)
38 Call objAttr.Put("showInAdvancedViewOnly", True)
39 Call objAttr.Put("searchFlags", 0)
40
41 Err.Clear
42 objAttr.SetInfo 'write the new attribute to the server
43 Call CheckIfError("create the attribute")
44 WScript.Echo "Created the attribute"
45
46 'Err.Clear
47 'Set objDSE = GetObject("LDAP://dc1.sanao.com/rootDSE")
48 'Call objDSE.Put("schemaUpdateNow", 1) 'trigger schema cache update
49 'objDSE.SetInfo
50 'Call CheckIfError("trigger schema cache update")
51 'WScript.Echo "Updated the schema cache"
52
53 Set objClass = objSchema.Create("classSchema", _
54     "CN=Sanao-Com-2000-MyProduct-Config-Info")
55
56 Call objClass.Put("LDAPDisplayName", _
57     "sanaoCom2000-MyProduct-ConfigInfo")
58 Call objClass.Put("governsID", "1.3.6.1.4.1.123123123.1.1")
59 Call objClass.Put("mustContain", _
60     Array("1.3.6.1.4.1.123123123.2.1"))
61 'Call objClass.Put("mustContain", _
62 '     Array("sanaoCom2000-MyProduct-UserData"))
63 Call objClass.Put("possSuperiors", Array("container"))
64 Call objClass.Put("objectClassCategory", 1) 'Structural
65 Call objClass.Put("subClassOf", "top")
66 Call objClass.Put("defaultHidingValue", True)
67 Call objClass.Put("showInAdvancedViewOnly", True)
68
69 Err.Clear
70 objClass.SetInfo 'write the new class to the server
71 Call CheckIfError("create the class")
72 WScript.Echo "Created the class"
73
74 Err.Clear
75 Set objDSE = GetObject("LDAP://dc1.sanao.com/rootDSE")
76 Call objDSE.Put("schemaUpdateNow", 1) 'trigger schema cache update

```

```
77 objDSE.SetInfo
78 Call CheckIfError("trigger schema cache update")
79 WScript.Echo "Updated the schema cache"
80
81 Sub CheckIfError(strAttemptedOperation)
82     If Err <> 0 Then
83         MsgBox "Couldn't " & strAttemptedOperation & vbCrLf & _
84             "Error code " & Hex(Err), _
85             vbOKOnly + vbCritical, "Error"
86         WScript.Quit(1) 'there was an error so we set the errorlevel
87     End If
88 End Sub
89
```